

C3M5a


Tangent Planes

We will learn soon that if f is a real-valued function with domain $D \subseteq \mathbb{R}^2$, i.e. $D \xrightarrow{f} \mathbb{R}$ then the gradient of f at $X = (x, y)$ denoted by $\nabla f(X)$, $grad f(X)$, or $grad f|_X$ is defined by

$$\nabla f(X) = \left\langle \frac{\partial f}{\partial x} \Big|_X, \frac{\partial f}{\partial y} \Big|_X \right\rangle = \langle f_x(x, y), f_y(x, y) \rangle = f_x(x, y)\vec{i} + f_y(x, y)\vec{j}$$

The gradient is a very important concept that is useful when discussing rates of change of functions of several variables. It is usually introduced in a course after partial derivatives have been defined, the chain rule has been discussed, and when the applications are being surveyed. Once the concept of the derivative of a real-valued multivariable function is introduced, the student can understand that this derivative somehow coincides with the gradient. The actual derivative at a point is a *linear mapping* which is evaluated by multiplication by a matrix. And that matrix is none other than the gradient (with the commas deleted).


In the single variable case where $y = f(x)$ and x_0 is a point in the domain, if $m = f'(x_0)$, then $\langle m \rangle$ is the gradient. The slope of the tangent line at x_0 is m and if one moves one unit to the right of x_0 , then the change in value for the tangent line is exactly m . So this simple diagram is valid.




Returning to the case where $z = f(x, y)$, the same geometric approach is valid. When (x_0, y_0) is in the domain of f and the graph of f is viewed in \mathbb{R}^3 , consider the vertical planes containing $(x_0, y_0, 0)$ that are parallel to the xz -plane and the yz -plane. Let's put this on a more concrete footing. Suppose that

$$\frac{\partial f}{\partial x} \Big|_{(x_0, y_0)} = f_x(x_0, y_0) = m_1 \quad \text{and} \quad \frac{\partial f}{\partial y} \Big|_{(x_0, y_0)} = f_y(x_0, y_0) = m_2$$

While it is true that $\nabla f(x_0, y_0) = \langle m_1, m_2 \rangle$, our focus here is not on the gradient. Rather, for now it is sufficient that one understand that we have evaluated the partial derivatives at (x_0, y_0) and obtained two **numbers** that serve as slopes in the x and y directions respectively. Then, we construct the vectors $\vec{v}_1 = \langle 1, 0, m_1 \rangle$ and $\vec{v}_2 = \langle 0, 1, m_2 \rangle$ and note that each is a **tangent vector** to the surface Σ defined by $z = f(x, y)$ when they are viewed as emanating from $\vec{X}_0 = (x_0, y_0, f(x_0, y_0))$. It is very important to have a geometric understanding of where \vec{v}_1 and \vec{v}_2 fit in the picture. We see that \vec{v}_1 is the tangent vector in the vertical plane parallel to the xz -plane and \vec{v}_2 is the tangent vector in the vertical plane parallel to the yz -plane. Each is aimed downward here. Observe the diagrams that follow.





For purposes of finding the basic equation for a plane tangent to the surface Σ , it is best to use $-\vec{N} = \vec{N}_1 = \langle m_1, m_2, -1 \rangle$. If $\vec{X} = \langle x, y, z \rangle$

$$\begin{aligned} \vec{N}_1 \cdot (\vec{X} - \vec{X}_0) &= 0 \\ \implies m_1(x - x_0) + m_2(y - y_0) - (z - z_0) &= 0 \\ \implies z - z_0 &= m_1(x - x_0) + m_2(y - y_0) \end{aligned}$$

yields three forms of our equation for the tangent plane. It is important to remember that m_1 and m_2 are numbers and are **NOT** expressions including variables.

Maple Example: For $f(x, y) = (9 - 2x^2 - y^2)/3$ and $(x_0, y_0) = (1, 1)$, find:

(a) a vector normal to the surface defined by $z = f(x, y)$ at $(1, 1)$, and an equation for the tangent plane at $X_0 = (x_0, y_0, f(x_0, y_0))$.

(b) Plot the surface, normal vector, tangent plane, and a line from $(x_0, y_0, 0)$ to X_0 .

```
> restart:      with(plots):      with(linalg):
> f:=(x,y)->(9-2*x^2-y^2)/3;
                                     f := (x, y) -> 3 - 2/3*x^2 - 1/3*y^2
> x0:=1;  y0:=1;  z0:=f(x0,y0);
                                     x0 := 1
                                     y0 := 1
                                     z0 := 2
> X:=[x,y,z];  X0:=[x0,y0,z0];
                                     X := [x, y, z]
                                     X0 := [1, 1, 2]
> fx:=diff(f(x,y),x);  fy:=diff(f(x,y),y);
                                     fx := -4/3*x
                                     fy := -2/3*y
> m1:=subs(x=x0,y=y0,fx);  m2:=subs(x=x0,y=y0,fy);
                                     m1 := -4/3
                                     m2 := -2/3
```

Our tangent vectors in the x and y directions are determined.

```
> vx:=[1,0,m1];  vy:=[0,1,m2];
```

$$vx := \left[1, 0, \frac{-4}{3} \right]$$

$$vy := \left[0, 1, \frac{-2}{3} \right]$$

We find the normal vector, which is orthogonal to our two tangent vectors.

```
> N:=crossprod(vx,vy);
```

$$N := \left[\frac{4}{3}, \frac{2}{3}, 1 \right]$$

```
> xaxis:=spacecurve([t,0,0],t=0..3,color=black):
```

```
> yaxis:=spacecurve([0,t,0],t=0..3,color=black):
```

```
> zaxis:=spacecurve([0,0,t],t=0..4,color=black):
```

```
> surff:=plot3d(f(x,y),x=0..2,y=0..2,color=cyan):
```

The line which represents the normal vector to the surface is plotted using a vector expression. We use 'evalm' to evaluate the expression to obtain the vector format we need for 'spacecurve'.

```
> Nline:=spacecurve(evalm(X0+t*N),t=0..1,color=magenta):
```

```
> X1:=vector([x0,y0,0]);
```

$$X1 := [x0, y0, 0]$$

Important: The easiest way to parameterize a line segment between two points (or vectors) P and Q is as $(1-t)P + tQ$ for $0 \leq t \leq 1$. We apply this in our next plot.

```
> Vline:=spacecurve(evalm((1-t)*X1+t*X0),t=0..1,color=blue):
```

Use the basic equation $\vec{N} \cdot \vec{X} = \vec{N} \cdot \vec{X}_0$ for a tangent plane to get an expression for z .

```
> eq1:=dotprod(N,X)=dotprod(N,X0);
```


$$eq1 := \frac{4}{3}x + \frac{2}{3}y + z = 4$$

```
> zee:=solve(eq1,z);
```

$$zee := -\frac{4}{3}x - \frac{2}{3}y + 4$$

```
> tplane:=plot3d(zee,x=.3..(1.7),y=.3..(1.7),color=sienna):
```

```
> display(xaxis,yaxis,zaxis,surff,Nline,Vline,tplane);
```



C3M5a Problem: Given $f(x, y) = 2 \cos(x) + 2 \sin(x) \cos(y)$ and $(x_0, y_0) = (\pi/3, -\pi/3)$. Use Maple to:

(a) Find an equation of the tangent plane to the surface $z = f(x, y)$ at $(x_0, y_0, f(x_0, y_0))$.

(b) Plot the surface for $0 \leq x \leq 1.8$ and $-1.5 \leq y \leq .5$, a line representing the normal vector at $(x_0, y_0, f(x_0, y_0))$, a line from $(x_0, y_0, 0)$ to $(x_0, y_0, f(x_0, y_0))$. Include the coordinate axes. Hint: let the

y -axis range from -2 to 2.